EPA-340/1-85-018

SOURCE TEST CALCULATION AND CHECK PROGRAMS FOR HEWLETT-PACKARD 41 CALCULATORS

Stationary Source Compliance Training Series

September 1985

EPA-340/1-85-013

SOURCE TEST CALCULATION AND CHECK PROGRAMS FOR HEWLETT-PACKARD 41 CALCULATORS

Stationary Source Compliance Training Series

September 1985

DISTRIBUTION RECORD

This document, Source Test Calculation and Check Programs for Hewlett-Packard 41 Calculators, has been prepared such that future additions and revisions can be added to update the information therein. A record of its distribution is being established and will be maintained up to date so that future revisions of existing material and additions may be distributed to document users.

In order to enter the document user's name and address in the distribution record, the "Distribution Record Card" below must be filled out and mailed to the address shown below. Any future change in name and/or address should also be sent to the same address.

In addition, document users who wish copies of the calculation and check programs on magnetic cards should send a HP program card holder and 40 program cards with a request to the same address.

Mr. Frank Clay
Emission Measurement Branch (MD-13)
U. S. Environmental Protection Agency
Research Triangle Park, NC 27711

(cut along dotted line)

DISTRIBUTION RECORD CARD

_____ Date _____
Last Name First Middle Initial

Address to Send Future Revisions and Additions

_____
Street

_____
City State Zip Code

If address is an employer or affiliate (fill in) _____

Employer or Affiliate

I have received a copy of Source Test Calculation and Check Programs for Hewlett-Packard 41 Calculators. Please send me any revisions and new additions to this volume of the Handbook.

EPA 340/1-85-018

Source Test Calculation And Check Programs For Hewlett-Packard 41 Calculators

Prepared By
Frank Clay

Emission Measurement Branch

U.S.ENVIRONMENTAL PROTECTION AGENCY
Emission Standards and Engineering Division
Office Of Air Quality Planning And Standards
Research Triangle Park, NC 27711

September 1985

ACKNOWLEDGEMENTS

This booklet was written by Mr. Frank Clay of the U. S. Environmental Protection Agency's Emission Measurement Branch (EMB), Emission Standards and Engineering Division. The programs herein were developed by Mr. Clay to assist the EMB and its contractors in their review and validation of emission measurement data in the field. It was edited and compiled for this printing by Entropy Environmentalists, Inc. For further information, contact Mr. Frank Clay at (919) 541-5543.

TABLE OF CONTENTS

INTRODUCTION

The programs presented in this booklet are used by the Emission Measurement Branch of the Office of Air Quality, Planning, and Standards, United States Environmental Protection Agency for checking field calculations, source test reports, or for in-house tests. The programs are written for the Hewlett-Packard 41-C, 41-CV, or the 41CX hand calculators. The programs can be modified to run on the HP-67 or HP-97, but the prompts would have to be eliminated and the memories reassigned since the earlier models have smaller memories. Since the Hewlett-Packard calculators use the RPN System, the programs would have to also have to be modified to run on calculators using the Algebraic Operating System common to such calculators as the TI-58 and TI-59.

Programs cover Methods 1 through 6, and there are two additional programs that can be used with the Alignment Method. A copy of each program is given along with some brief instructions on how to use it. Occasional examples are also included..

METER BOX PROGRAM (DELTA H)

The meter box program is used to set the Delta H of the meter box during the sample run. The program is more accurate and easier to use than the nomograph. Unlike the nomograph, nothing in the program needs to be reset if there is a significant change in stack temperature. Initial prompts for data input are as follows:

(1) Number of thermometers in the dry gas meter
(2) Diameter of Nozzle
(3) Delta $H_@$
(4) Pitot Coefficient
(5) Per Cent moisture
(6) Barometric Pressure (Inches Mercury)
(7) Static Pressure of Stack (Inches Water)
(8) Molecular Weight Dry
(9) Molecular Weight Wet

After the above data are entered, the program goes into a short loop and prompts for the following data for each sample point:

(10) Meter inlet temperature (^{o}F)
(11) Meter outlet temperature (^{o}F)
(12) Stack temperature (^{o}F)
(13) Stack Delta p (Inches water)

Note that if the dry gas meter has only one temperature indicator, prompts 9 and 10 will be replaced by a single prompt: METER TEMP?

After entering the above data, the Delta H setting for the meter box will appear in the display. After the Delta H is determined, pressing R/S will reset the program for the next sample point. The prompt that will appear in the display will be MTR IN TEMP? if the dry gas meter has two thermometers, or METER TEMP? if the dry gas meter has only one temperature gauge.

When all the sample points have been completed, there is a subroutine, Subroutine A, that can be manually executed to give the average meter temperature, average stack temperature, average Delta H, and the average of the individual square roots of $[\text{Delta } p_s \ (T_s + 460)]$.

SPECIFIC PROBLEMS RUNNING THE METER BOX

There are some problems that may occur during the course of a run. First, suppose a mistake is made during the input of the first nine pieces of data. To correct the mistake, simply go to the beginning of the program and re-enter all data via prompts.

Next, suppose a mistake is made entering data while taking the sample (steps 10 to 13). (A mistake is when a piece of incorrect data is entered and R/S is pressed.) To correct this error, simply press R/S until you reach the meter temperature prompt. This will be either the MTR IN TEMP? prompt if the meter box has two temperature indicators, or the METER TEMP? prompt if the meter box has a single temperature gauge. When the prompt is reached, re-enter the correct pieces of data for that sample point.

Another problem that can occur during a test is a process interruption. This means that the train must be shut down until the process is re-started. When the test is resumed, it is not likely that stack conditions will be identical to those when the test was stopped, so another Delta H will have to be calculated. Simply treat the re-start as another point for calculation purposes. (Note: data reduction will have to be time weighted unless the train was stopped at the exact end of a point.)

When the test run is concluded, one of two situations will exist:

(1) The test went perfectly, that is, there were no process interruptions and no data were entered incorrectly during the run. In this case, Subroutine A will provide accurate summaries of the average stack temperature, meter temperature, Delta H, and $[\text{Delta } p_s\ (T_s + 460)]^{1/2}$.

(2) The test had a process interruption or some data were entered incorrectly in the Delta H loop. If a process interruption occurred, and no mistakes were made entering data for the Delta H calculations, Subroutine A will give a reasonable approximation of the average values needed for isokinetic calculation although the final numbers will have to be time weighted for preciseness (A program labeled TWA in the back of this booklet may be used to determine time weighted averages.) If an error occurred while inputting data for the Delta H calculation and R/S was pressed through the loop, disregard Subroutine A.

One last note - any HP-41 calculator will automatically turn itself off after ten minutes of inactivity. This means that during port changes and process interruptions, the display would be blank when the test is resumed. Executing the ON function will override this feature. To execute the ON function, simply press

XEQ ALPHA ON ALPHA. The calculator will remain on until it is manually turned off. Turning the calculator off automatically clears the ON function.

The label for this program is DELTA H. The Memory size is 019.

```
          XEQ "DELTA H"
NO. OF THERMOMETERS ON
DRY GAS METER?
            2.00    RUN
DIA NOZZLE?
             .26    RUN
DELTA Ha ?
            1.84    RUN
PITOT CP ?
             .84    RUN
% MOISTURE?
           13.50    RUN
BAR PRESS?
           29.55    RUN
STATIC H2O?
            -.38    RUN
MOL WT DRY?
           28.95    RUN
MOL WT WET?
           27.47    RUN
MTR IN TEMP?
           82.00    RUN
MTR OUT TEMP?
           73.00    RUN
STACK TEMP?
          123.00    RUN
DELTA P?
             .21    RUN
----------
*DELTA H = 0.75
                    RUN

MTR IN TEMP?
           82.00    RUN
MTR OUT TEMP?
           73.00    RUN
STACK TEMP?
          123.00    RUN
DELTA P?
             .21    RUN
----------
*DELTA H = 0.75
                    RUN

MTR IN TEMP?
           82.00    RUN
MTR OUT TEMP?
           73.00    RUN
STACK TEMP?
          125.00    RUN
DELTA P?
             .22    RUN
----------
*DELTA H = 0.80
                    RUN

MTR IN TEMP?
           82.00    RUN
MTR OUT TEMP?
           73.00    RUN
STACK TEMP?
          128.00    RUN
DELTA P?
             .37    RUN
----------
*DELTA H = 1.33
                    RUN

MTR IN TEMP?
           83.00    RUN
MTR OUT TEMP?
           74.00    RUN
STACK TEMP?
          127.00    RUN
DELTA P?
             .36    RUN
----------
*DELTA H = 1.30
                    RUN

MTR IN TEMP?
           84.00    RUN
MTR OUT TEMP?
           74.00    RUN
STACK TEMP?
          127.00    RUN
DELTA P?
             .40    RUN
----------
*DELTA H = 1.45
                    RUN

MTR IN TEMP?
           85.00    RUN
MTR OUT TEMP?
           74.00    RUN
STACK TEMP?
          127.00    RUN
DELTA P?
             .39    RUN
----------
*DELTA H = 1.41
                    RUN

MTR IN TEMP?
           85.00    RUN
MTR OUT TEMP?
           74.00    RUN
STACK TEMP?
          127.00    RUN
DELTA P?
             .40    RUN
----------
*DELTA H = 1.45
                  SF 05
                    RUN

METER TEMP ?
           80.50    RUN
STACK TEMP?
          127.00    RUN
DELTA P?
             .36    RUN
----------
*DELTA H = 1.31
                    RUN

METER TEMP ?
           80.50    RUN
STACK TEMP?
          126.00    RUN
DELTA P?
             .35    RUN
----------
*DELTA H = 1.27
                    RUN

METER TEMP ?
           80.50    RUN
STACK TEMP?
          127.00    RUN
DELTA P?
             .27    RUN
----------
*DELTA H = 0.98
                    RUN

METER TEMP ?
           81.00    RUN
STACK TEMP?
          127.00    RUN
DELTA P?
             .20    RUN
----------
*DELTA H = 0.73
                  XEQ A
TOTAL POINTS?
           12.00    RUN
AVE MTR TEMP = 79.
AVE STK TEMP = 126.
AVE DELTA H = 1.13
SQRT PSTS = 13.4817
```

Note that Flag 05 was set. From this point on, only one meter temperature prompt occurred.

```
PRP "DELTA H
          "

 01*LBL "DEL
        TA H"
 02 FIX 2
 03 CF 05
 04 CLRG
 05 1
 06 "NO. OF
   THERMOME"
 07 "HTERS O
          N"
 08 AVIEW
 09 "DRY GAS
     METER?"
 10 PROMPT
 11 X=Y?
 12 SF 05
 13 "DIA NOZ
        ZLE?"
 14 PROMPT
 15 4
 16 Y↑X
 17 "DELTA H
        a ?"
 18 PROMPT
 19 STO 04
 20 *
 21 846.872
 22 *
 23 "PITOT C
        P ?"
 24 PROMPT
 25 X↑2
 26 *
 27 STO 05
 28 "% MOIST
        URE?"
 29 PROMPT
 30 100
 31 /
 32 1
 33 -
 34 X↑2
 35 RCL 05
 36 *
 37 STO 06
 38 "BAR PRE
        SS?"
 39 PROMPT
 40 STO 07
 41 "STATIC
        HOH?"
 42 PROMPT
 43 13.6
 44 /
 45 RCL 07
 46 +
 47 "MOL WT
        DRY?"
 48 PROMPT
 49 *
 50 "MOL WT
        WET?"
 51 PROMPT
 52 /
 53 STO 08
 54 RCL 04
 55 13.6
 56 /
 57 RCL 07
 58 +
 59 RCL 08
 60 X<>Y
 61 /
 62 RCL 06
 63 *
 64 STO 09

 65*LBL 01
 66 FS? 05
 67 GTO 02
 68 "MTR IN
      TEMP?"
 69 PROMPT
 70 ENTER↑
 71 "MTR OUT
      TEMP?"
 72 PROMPT
 73 +
 74 2
 75 /
 76 ST+ 11
 77 460
 78 +
 79 STO 10
 80 FC? 05
 81 GTO 03

 82*LBL 02
 83 "METER T
       EMP ?"
 84 PROMPT
 85 ST+ 11
 86 460
 87 +
 88 STO 10

 89*LBL 03
 90 "STACK T
       EMP?"
 91 PROMPT
 92 ST+ 12
 93 460
 94 +
 95 STO 15
 96 RCL 10
 97 X<>Y
 98 /
 99 RCL 09
100 *
101 "DELTA P
          ?"
102 PROMPT
103 STO 16
104 *
105 ST+ 13
106 STO 18
107 RCL 15
108 RCL 16
109 *
110 SQRT
111 ST+ 17
112 RCL 18
113 "-------
          ---"
114 AVIEW
115 "*DELTA
       H = "
116 ARCL X
117 AVIEW
118 STOP
119 FC? 55
120 GTO 01
121 ADV
122 GTO 01

123*LBL A
124 "TOTAL P
      OINTS?"
125 PROMPT
126 STO 14
127 RCL 11
128 X<>Y
129 /
130 FIX 0
131 "AVE MTR
     TEMP = "
132 ARCL X
133 AVIEW
134 FC? 55
135 STOP
136 RCL 12
137 RCL 14
138 /
139 "AVE STK
     TEMP = "
140 ARCL X
141 AVIEW
```

```
142 FC? 55
143 STOP
144 RCL 13
145 RCL 14
146 /
147 FIX 2
148 "AVE DEL
    TA H = "
149 ARCL X
150 AVIEW
151 FC? 55
152 STOP
153 RCL 17
154 RCL 14
155 /
156 FIX 4
157 "SQRT PS
      TS = "
158 ARCL X
159 AVIEW
160 .END.
```

STACK GAS ANALYSIS (METH 3)

This program determines the molecular weight of the stack gas. It can determine the dry gas weight only, or it can determine both wet and dry molecular weights. If another gas is present in the gas stream - such as SO_2, this can be added and the molecular weight of the total stack effluent determined. If the calculator "prompts" for data none exists, (such as CO) simply press R/S and continue.

The program uses flags for different run configurations. The simpliest way to use the flags is to set no flags for dry molecular weight only, and to set Flag 00 to get both wet and dry molecular weights. In either case, some prompts will appear that are not needed, but pressing R/S in these instances will cause the program to continue. If eliminating unnecessary prompts is desirable, the table below gives the flag settings for different gas components.

DRY MOLECULAR WEIGHT ONLY

Stack Gas Components	Set Flag(s)
CO_2 and O_2	00, 01, 02, 04, and 08
CO_2, O_2, and CO	00, 01, 02, and 04
CO_2, O_2, CO, and another gas	00, 01, and 02

WET AND DRY MOLECULAR WEIGHTS

H_2O, CO_2, and O_2	00, 04, and 08
H_2O, CO_2, O_2, and CO	00 and 04
H_2O, CO_2, O_2, CO, and another gas	00

The label for this program is METH 3. The Memory size is 045.

```
PRP "METH 3"

 01*LBL "MET
        H 3"
 02 FS? 00
 03 CLRG
 04 FS? 01
 05 GTO 01
 06 RCL 20
 07 X≠0?
 08 GTO 01
 09 RCL 22
 10 X=0?
 11 GTO A
 12 GTO 01

 13*LBL A
 14 "% MOIST
      URE ?"
 15 PROMPT
 16 STO 22

 17*LBL 01
 18 ADV
 19 0
 20 "% CO2?"
 21 PROMPT
 22 STO 24
 23 "% OXYGE
       N?"
 24 PROMPT
 25 STO 25
 26 0
 27 FS? 00
 28 GTO 98
 29 "% CO ?"
 30 PROMPT

 31*LBL 98
 32 STO 27
 33 0
 34 FS? 04
 35 GTO 99
 36 "MOL WT
    OTHER?"
 37 PROMPT

 38*LBL 99
 39 .01
 40 *
 41 STO 28
 42 X=0?
 43 GTO 02
 44 RCL 09
 45 X=0?
 46 GTO 03
 47 10000
 48 *
 49 STO 42
 50 GTO 02

 51*LBL 03
 52 0
 53 "PPM ?"
 54 PROMPT
 55 STO 42

 56*LBL 02
 57 RCL 24
 58 RCL 25
 59 +
 60 RCL 27
 61 +
 62 RCL 42
 63 .0001
 64 *
 65 STO 43
 66 +
 67 100
 68 -
 69 CHS
 70 STO 26
 71 RCL 24
 72 .44
 73 *
 74 STO 29
 75 RCL 25
 76 .32
 77 *
 78 ST+ 29
 79 RCL 26
 80 RCL 27
 81 +
 82 .28
 83 *
 84 ST+ 29
 85 RCL 43
 86 RCL 28
 87 *
 88 ST+ 29
 89 STO 44
 90 RCL 29
 91 ADV
 92 FIX 2
 93 "MWd ="
 94 ARCL X
 95 AVIEW
 96 PSE
 97 FS? 02
 98 STOP
 99 100
100 RCL 22
101 -
102 100
103 /
104 STO 23
105 1
106 RCL 23
107 -
108 18
109 *
110 RCL 29
111 RCL 23
112 *
113 +
114 STO 30
115 "MW WET=
                "
116 ARCL X
117 AVIEW
118 ADV
119 FIX 4
120 RTN
121 .END.
```

MOISTURE CONTENT DETERMINATION (METH 4)

The Method 4 program calculates moisture content. It will also check for saturation (as might be the case at a wet scrubber outlet) and it will calculate the moisture content of a gas stream if a gas such as SO_2 is removed by peroxide impingers before reaching the dry gas meter. Flags are used to check for saturation of the gas stream, or to correct for a gas removed before the dry gas meter.

One thing that the program does not do is to make the distinction between grams of H_2O and ml. of H_2O. The Federal Register determines moisture content by multiplying the impinger ml. of water by 0.04707 to get standard cubic feet of H_2O and grams of H_2O in the silica gel impinger by 0.04715 to get standard cubic feet of H_2O from the impinger. This is because there is a very slight difference between a gram of H_2O and a ml. of H_2O. The Method 4 program assumes that grams and ml. are the same, thus, the total of ml. + grams is used for calculations and multiplied by 0.04707.

Flags for the program are used as follows:

(a) Flags 3 and 4 are set for most operations. This is when no gas was removed by peroxide impingers and stack gas saturation is not a problem

(b) Flag 3 is set. A gas was removed from the sample before reaching the dry gas meter (SO_2 is an example). The calculator will prompt for the per cent of the gas removed.

(c) Flag 04 is set. Stack gas saturation is possible and the water collected in the impingers may give a higher than saturation moisture content. The calculator will prompt for the vapor pressure of the stack gas at stack temperature. (Obtain the vapor pressure from the chart that follows the METH 4 program.)

(d) No Flags set. A gas was removed from the gas stream (such as SO_2) before the dry gas meter and the stack gas was checked for saturation.

The flags in the program are used to eliminate or by-pass part of the program. When Flag 03 is set, the program does not prompt for Vapor Pressure. When Flag 04 is set, the program does not prompt for a gas removed before the dry gas

meter. With no flags set, the program prompts for both "saturation" and "other gas removed."

If you forget to set the flag(s) and the calculator prompts for data input but none is needed, simply press R/S and continue. The program will just take a little longer to run.

The label for this program is METH 4. The Memory size is 045.

```
 01*LBL "MET
H 4"
 02 CLRG
 03 1
 04 STO 42
 05 "METER B
OX Y?"
 06 PROMPT
 07 STO 03
 08 "DELTA H
?"
 09 PROMPT
 10 STO 04
 11 "BAR PRE
SS ?"
 12 PROMPT
 13 STO 05
 14 "METER V
OL ?"
 15 PROMPT
 16 STO 06
 17 "MTR TEM
P F?"
 18 PROMPT
 19 STO 07
 20 0
 21 FS? 04
 22 GTO 99
 23 BEEP
 24 "% OTHER
 GAS"
 25 AVIEW
 26 "REMOVED
 BEFORE"
 27 AVIEW
 28 "DRY GAS
 METER ?"
 29 PROMPT
 30*LBL 99
 31 STO 09
 32 "STATIC
HOH IN ?"
 33 PROMPT
 34 STO 10
 35 "STACK T
EMP."
 36 PROMPT
 37 STO 18
 38 FS? 03
 39 GTO 98
 40 BEEP
 41 "VAPOR P
R IN HG?"
 42 PROMPT
 43*LBL 98
 44 STO 19
 45 0
 46 RCL 20
 47 "ML. WAT
ER ?"
 48 PROMPT
 49 STO 20
 50 X≠0?
 51 GTO 01
 52 100
 53 "% MOIST
URE ?"
 54 PROMPT
 55 STO 22
 56*LBL 01
 57 RCL 04
 58 13.6
 59 /
 60 RCL 05
 61 +
 62 STO 17
 63 RCL 06
 64 *
 65 RCL 03
 66 *
 67 17.647
 68 *
 69 RCL 07
 70 460
 71 +
 72 /
 73 STO 08
 74 100
 75 RCL 09
 76 X=0?
 77 GTO 07
 78 -
 79 .01
 80 *
 81 RCL 08
 82 X<>Y
 83 /
 84 STO 08
 85*LBL 07
 86 RCL 20
 87 X=0?
 88 GTO 02
 89 RCL 20
 90 .0471
 91 *
 92 STO 21
 93 100
 94 *
 95 RCL 08
 96 RCL 21
 97 +
 98 /
 99 STO 22
100 STO 13
101*LBL 02
102 RCL 10
103 13.6
104 /
105 RCL 05
106 +
107 STO 17
108 RCL 19
109 X<>Y
110 /
111 100
112 *
113 ADV
114 ADV
115 STO 12
116 100
117 X<Y?
118 GTO 05
119 GTO 06
120*LBL 06
121 RCL 12
122 FIX 1
123 "SAT % =
 "
124 ARCL X
125 AVIEW
126 FIX 4
127*LBL 05
128 RCL 20
129 X=0?
130 GTO 04
131 RCL 22
132*LBL 04
133 ADV
134 ADV
135 FIX 1
136 "IMP. %
HOH = "
137 ARCL X
138 AVIEW
139 FIX 4
140 RCL 12
141 RCL 22
142 X<=Y?
143 GTO 03
144 RCL 12
145*LBL 03
146 STO 22
147 ADV
148 FIX 1
149 "% HOH="
150 ARCL X
151 AVIEW
152 ADV
153 FIX 4
154 END
```

This chart may be used with the Method 4 program when checking the stack gas for saturation. The chart gives the vapor pressure of water in whole Degrees Fahrenheit. The equation for stack gas saturation is:

$$\% \text{ Saturation Moisture Content} = \frac{\text{Vapor Pressure HOH @ Stack Temp. X 100}}{\text{Absolute Stack Pressure}}$$

Rnn equals the temperature in Degrees F. Thus R66 = 0.6441 equals the vapor pressure of water at 66° Fahrenheit.

R32=	0.3363	R71=	0.7648	R111=	2.673
R33=	0.3491	R72=	0.7911	R112=	2.751
R34=	0.3624	R73=	0.8183	R113=	2.831
R35=	0.3761	R74=	0.8463	R114=	2.913
R36=	0.3903	R75=	0.8751	R115=	7.576
R37=	0.4049	R76=	0.9047	R116=	3.082
R38=	0.4200	R77=	0.9352	R117=	3.170
R39=	0.4356	R78=	0.9667	R118=	3.261
R40=	0.4518	R79=	0.9990	R119=	3.353
R41=	0.4685	R80=	1.0323	R120=	3.448
R42=	0.4856	R81=	1.0665	R121=	3.545
R43=	0.5033	R82=	1.1017	R122=	3.644
R44=	0.5216	R83=	1.1380	R123=	3.746
R45=	0.5405	R84=	1.1752	R124=	3.850
R46=	0.5599	R85=	1.1236	R125=	3.956
R47=	0.5800	R86=	1.2530	R126=	4.065
R48=	0.3363	R87=	1.2935	R127=	4.177
R49=	0.3491	R88=	1.3351	R128=	4.291
R50=	0.3624	R89=	1.3779	R129=	4.408
R51=	0.3761	R90=	1.4219	R130=	4.527
R52=	0.3903	R91=	1.4671	R131=	4.650
R53=	0.4049	R92=	1.5136	R132=	4.775
R54=	0.4200	R93=	1.5613	R133=	4.903
R55=	0.4356	R94=	1.6103	R134=	5.034
R56=	0.4518	R95=	1.6607	R135=	5.168
R57=	0.4685	R96=	1.7124	R136=	5.305
R58=	0.4856	R97=	1.7655	R137=	5.445
R59=	0.5033	R98=	1.8200	R138=	5.588
R60=	0.5216	R99=	1.8759	R139=	5.735
R61=	0.5405	R100=	1.933	R140=	5.884
R62=	0.5599	R101=	1.992	R141=	6.037
R63=	0.5800	R102=	2.053	R142=	6.193
R64=	0.6007	R103=	2.115	R143=	6.353
R65=	0.6221	R104=	2.179	R144=	6.516
R66=	0.6441	R105=	2.244	R145=	6.683
R67=	0.6668	R106=	2.311	R146=	6.854
R68=	0.6902	R107=	2.380	R147=	7.028
R69=	0.7143	R108=	2.450	R148=	7.206
R70=	0.7392	R109=	2.523	R149=	7.387
		R110=	2.673	R150=	7.573

R151=	7.762
R152=	7.956
R153=	8.153
R154=	8.355
R155=	8.561
R156=	8.771
R157=	8.985
R158=	9.204
R159=	9.428
R160=	9.656
R161=	9.888
R162=	10.13
R163=	10.37
R164=	10.62
R165=	10.87
R166=	11.12
R167=	11.39
R168=	11.65
R169=	11.93
R170=	12.20
R171=	12.49
R172=	12.78
R173=	13.07
R174=	13.37
R175=	13.68
R176=	13.99
R177=	14.31
R178=	14.63
R179=	14.96
R180=	15.30
R181=	15.64
R182=	15.99
R183=	16.34
R184=	16.70
R185=	17.07
R186=	17.45
R187=	17.83
R188=	18.22
R189=	18.61
R190=	19.02
R191=	19.43
R192=	19.85
R193=	20.27
R194=	20.70
R195=	21.15
R196=	21.59
R197=	22.05
R198=	22.52
R199=	22.99
R200=	23.47
R201=	23.96
R202=	24.46
R203=	24.96
R204=	25.48
R205=	26.00
R206=	26.53
R207=	27.07
R208=	27.63
R209=	28.19
R210=	28.75
R211=	29.33
R212=	29.92

NOZZLE SELECTION PROGRAM (NOZZLE)

This program replaces the Nomograph for determining the nozzle size. It can select a nozzle size based on the standard sampling rate of 0.75 dry standard cubic feet per minute just as the Nomograph does, or Flag 00 may be set and any desired sampling rate may be entered. The program requires the following inputs:

Sample Rate Fixed at 0.75 DSCFM (No Flags Set)	Sample Rate to be Choosen (Set Flag 00)
Stack Static Pressure, In. H_2O	Stack Static Pressure In. H_2O
Barometric Pressure In. Hg.	Barometric Pressure In. Hg.
Stack Temperature $^{o}F.$	Stack Temperature $^{o}F.$
Per Cent Moisture	Per Cent Moisture
Stack Velocity in FPS or FPM	Stack Velocity in FPS or FPM
	Desired Sample Rate

The label for this program is NOZZLE. The Memory size is 003.

```
PRP "NOZZLE"

 01*LBL "NOZ
        ZLE"
 02 CLRG
 03 FIX 2
 04 0

 05*LBL 00
 06 "STATIC
      HOH ?"
 07 PROMPT
 08 13.6
 09 /
 10 "BAR PRE
       SS ?"
 11 PROMPT
 12 +
 13 STO 00
 14 "STK TEM
       P ?"
 15 PROMPT
 16 460
 17 +
 18 STO 01
 19 0
 20 "% HOH ?
           "
 21 PROMPT
 22 100
 23 /
 24 1
 25 +
 26 RCL 01
 27 *
 28 FS? 00
 29 GTO c
 30 .0425

 31*LBL 01
 32 *
 33 RCL 00
 34 -
 35 STO 02
 36 0
 37 "STACK F
       PS ?"
 38 PROMPT
 39 X=0?
 40 GTO a
 41 60
 42 *
 43 GTO b

 44*LBL a
 45 "STACK F
       PM ?"
 46 PROMPT

 47*LBL b
 48 RCL 02
 49 X<>Y
 50 /
 51 PI
 52 /
 53 SQRT
 54 24
 55 *
 56 FIX 4
 57 "ACT. DI
      A. = "
 58 ARCL X
 59 AVIEW
 60 STOP
 61 GTO 00

 62*LBL c
 63 "MTR DSC
       FM ?"
 64 PROMPT
 65 .056667
 66 *
 67 GTO 01
 68 .END.
```

SAMPLING SUMMARY SHEET

Data from the field data sheets are recorded on the Sampling Summary Sheet. A number is found above the column abbreviations for the various values on the sheet. The number is the same as the calculator memory number in which the value is stored. If the number has a circle around it, it is a value that will be prompted for in the program. If there is no circle around the number, it is a value to be calculated by the program. A broken circle around the number indicates that it may or may not be used in the calculations. Two of the numbers have a diamond around them. These two values are used on the opposite side of the sheet for the summary of emissions calculations.

SUMMARY OF EMISSIONS SHEET

The Summary of Emissions Sheet is similar to the Sampling Summary Sheet. The top row has four circled numbers on it; two of the calues have already been calculated (the columns with the numbers surrounded by the diamonds on the preceeding page). The memory numbers for the Concentrations Section are found at the bottom of the columns instead of above them. These are all determined by the MASSFLO program.

Once the labratory analysis of the samples is completed and the milligrams of catch are determined, the emission rates for the front half, back half, and total catch can be obtained from the MASSFLO program.

SAMPLING SUMMARY SHEET

PLANT _____ LOCATION _____

SAMPLED SOURCE _____

Run	Date	N_p	(3) Y_m	(4) ΔH	(5) P Bar.	(6) V_m	(7) T_m	(8) V_m Std.	(9) %	(10) P_{st}	17 P_s	(18) T_s	(19) VP	(20) V_w	21 $V_{w_{gas}}$	22 %M	23 M_d

Run	(24) CO_2	(25) O_2	26 N_2	(27) CO	(28) ___	29 MW_d	30 MW	(31) $C_p\sqrt{\Delta P_s(T_s + 460)}$	(32)	(33) T_t	(34) D_n	35 V_s	(36) D_s	(37) Area	38 ACFM	39 DSCFM	40 %I

$$V_{m\ Std.} = Y_m \frac{17.64 \times V_m \left(P\ Bar. + \frac{\Delta H}{13.6}\right)}{(T_m + 460)}$$

$$V_{w_{gas}} = 0.0471\ V_w$$

$$M_d = \%CO_2 \times .44 + (\%O_2 \times .32) + (\%CO + \%N_2) \times .28 + \left(\%Other \times \frac{Mol.\ wt.}{100}\right)$$

$$M = MW_d \times M_d + 18(1 - M_d)$$

$$V_s = 85.49 \times C_p \sqrt{P_s (T_s + 460)} \left[\frac{1}{P_s \times MW}\right]^{1/2}$$

$$\%I = \frac{0.09450 \times (T_s + 460) \times V_{m\ Std.}}{V_s \times T_t \times P_s \times M_d \times A_n}$$

$$\%H = \frac{100 \times V_{w_{gas}}}{V_{m\ Std.} + V_{w_{gas}}}$$

$$M_d = \frac{100 - \%H}{100}$$

$$P_s = P\ Bar. \pm \frac{P_{st}}{13.6}$$

N_p = Total No. of Sampling points

Y_m = Meter Box Correction Factor

ΔH = Average Orfice Pressure Drop, Inches H_2O

P Bar. = Barometric Pressure, Inches Hg, Absolute

V_m = Volume of Dry Gas at Meter Conditions, DCF

T_m = Average Meter Temperature, ^oF.

$V_{m\ Std.}$ = Volume of Dry Gas at STP, DSCF[a]

% = Per Cent other gas removed before Dry Gas Meter

P_{st} = Static Pressure of Stack Gas, Inches H_2O

P_s = Stack Gas Pressure, Inches Hg.

T_s = Average Stack Temperature, ^oF

VP = Vapor Pressure of H_2O At Stack Temperature

V_w = Total H_2O Collected in Impingers and Silica Gel

$V_{w_{gas}}$ = Volume of water vapor collected at STP, SCF[b]

%H = Per Cent Moisture by volume

M_d = Mole Fraction of Dry Gas

$\%CO_2$ = Volume % Dry

$\%O_2$ = Volume % Dry

$\%N_2$ = Volume % Dry

%CO = Volume % Dry

MW_d = Molecular Weight of Stack Gas Dry Basis

MW = Molecular Weight of Stack Gas, Wet Basis

C_p = Pitot Tube Coefficient

$\sqrt{\Delta P (T_s + 460)}$ Is determined by averaging the square root of the product of the velocity head (ΔP) and the absolute stack temperature from each sampling point.

T_t = Net time of test in minutes

D_n = Sampling Nozzle Diameter, Inches

A_n = Area of Nozzle opening, ft^2.

V_s = Stack Gas Velocity at Stack Conditions, Feet per second.

D_s = Diameter of Stack, Inches

Area = Area of duct in ft^2

ACFM = Actual Cubic Feet per minute

DSCFM = Dry Standard Cubic Feet per minute

%I = Per Cent Isokinetic

[a] Dry Standard Cubic Feet @ 68°F, 29.92 in. Hg.

[b] Standard Conditions @ 68°F, 29.92 in. Hg.

PLANT _____ LOCATION _____

SAMPLED SOURCE _____

RUN	DATE	(1) V_m Std.	(2) DSCFM

FRONT HALF MILLIGRAMS			
PROBE	CYCLONE	FILTER	(3) TOTAL

(4) BACK HALF	5 FRONT + BACK TOTAL

CONCENTRATIONS

	FRONT HALF			
RUN	Gr/DSCF	Mg/M^3	Lb/Hr	Kg/Hr
	6	7	8	9

	BACK HALF			
Gr/DSCF	Mg/M^3	Lb/Hr	Kg/Hr	
10	17	18	19	

	TOTAL			
Gr/DSCF	Mg/M^3	Lb/Hr	Kg/Hr	
20	21	22	23	

Remarks: _____

ISOKINETIC DETERMINATION (METH 5)

The Method 5 program also runs the Method 3 and Method 4 programs. Since it runs the Method 4 program, the flags used in the METH 4 program can also be used to eliminate the prompt for % other gas removed before the dry gas meter, and the prompt for the vapor pressure of the stack gas at stack temperature. Forgetting to set the flags will just cause the program to prompt for data not needed in most cases. Simply press R/S and continue. The program will just take a little longer to run. For practical purposes, the flags are of benefit when calculating the isokinetic rate for a large number of runs since parts of the program can be skipped. Flag 04 is useful when saturation may be a problem.

Use the flags as follows:

(a) For a run where no gas was removed before the dry gas meter but saturation may occur, set Flag 04.

(b) For a run where no gas was removed before the dry gas meter and saturation is not a problem, set Flag 03.

(c) For a run where no gas was removed before the dry gas meter and saturation is not a problem, set Flags 03 and 04.

(d) Setting no flags will prompt for other gas removed before the dry gas meter and also for vapor pressure in the case of possible saturation.

(e) If the pitot coefficient is other than 0.84, set flag 06 and enter the pitot coefficient when prompted for.

(f) If sampling using the Alignment Method, set Flag 07.

(g) If using a Fyrite for stack gas composition, set Flag 08.

To determine the isokinetic rate using an assumed moisture content, do not enter anything for the ML WATER prompt - simply press R/S and the calculator will then ask for a moisture content.

When sampling a rectangular duct and the STK DIA INCHES? prompt appears, simply press R/S and the calculator will then prompt for AREA SQ FT? Enter the stack area.

Note that when a gas such as SO_2 is removed before the dry gas meter, the pounds per hour emission rate is also calculated.

The label for this program is METH 5. The Memory size is 045.

```
PRP "METH 5"

 01*LBL "MET
        H 5"
 02 FIX 4
 03 "RUN NUM
        BER"
 04 PROMPT
 05 XEQ "MET
        H 4"
 06 XEQ "MET
        H 3"
 07 ADV
 08 .84
 09 "PITOT C
        P ?"
 10 FS? 06
 11 PROMPT
 12 STO 31
 13 "SQRT PS
       TS ?"
 14 PROMPT
 15 STO 32
 16 FS? 07
 17 GTO a
 18 GTO b

 19*LBL a
 20 "COS SQR
   T PSTS ?"
 21 PROMPT
 22 STO 42

 23*LBL b
 24 "TIME MI
        N ?"
 25 PROMPT
 26 STO 33
 27 "NOZZLE
      DIA ?"
 28 PROMPT
 29 STO 34
 30 0
 31 "STK DIA
     INCH ?"
 32 PROMPT
 33 STO 36
 34 X=0?
 35 GTO 01
 36 "AREA SQ
       FT ?"
 37 PROMPT
 38 STO 37
 39 GTO 02

 40*LBL 01
 41 24
 42 /
 43 X↑2
 44 PI
 45 *
 46 STO 37

 47*LBL 02
 48 100
 49 RCL 22
 50 -
 51 100
 52 /
 53 STO 23
 54 1
 55 RCL 23
 56 -
 57 18
 58 *
 59 RCL 29
 60 RCL 23
 61 *
 62 +
 63 STO 30
 64 RCL 10
 65 13.6
 66 /
 67 RCL 05
 68 +
 69 STO 17
 70 1
 71 RCL 17
 72 /
 73 RCL 30
 74 /
 75 SQRT
 76 FS? 07
 77 GTO c
 78 GTO d

 79*LBL c
 80 RCL 43
 81 GTO e

 82*LBL d
 83 RCL 32

 84*LBL e
 85 *
 86 RCL 31
 87 *
 88 85.49
 89 *
 90 STO 35
 91 FS? 07
 92 GTO A

 93*LBL B
 94 RCL 34
 95 2
 96 /
 97 X↑2
 98 PI
 99 *
100 144
101 /
102 STO 41
103 RCL 18
104 460
105 +
106 RCL 08
107 *
108 .09450
109 *
110 FS? 07
111 RCL 43
112 FC? 07
113 RCL 35
114 /
115 RCL 33
116 /
117 RCL 17
118 /
119 RCL 23
120 /
121 RCL 41
122 /
123 STO 40
124 RCL 37
125 RCL 35
126 *
127 60
128 *
129 STO 38
130 RCL 23
131 *
132 RCL 17
133 29.92
134 /
135 *
136 RCL 18
137 460
138 +
139 528
140 X<>Y
141 /
142 *
143 STO 39
144 ADV
145 FIX 3
146 RCL 08
147 "* VOL M
    TR STD ="
148 "⊢ "
```

```
149 ARCL X
150 AVIEW
151 FIX 2
152 RCL 17
153 "  STK P
   RES ABS "
154 "⊢= "
155 ARCL X
156 AVIEW
157 RCL 21
158 "  VOL H
   OH GAS ="
159 "⊢ "
160 ARCL X
161 AVIEW
162 RCL 22
163 "  % MOI
   STURE = "
164 ARCL X
165 AVIEW
166 FIX 3
167 RCL 23
168 "  MOL D
   RY GAS ="
169 "⊢ "
170 ARCL X
171 AVIEW
172 FIX 2
173 RCL 26
174 "  % NIT
   ROGEN = "
175 ARCL X
176 AVIEW
177 RCL 29
178 "  MOL W
   T DRY = "
179 ARCL X
180 AVIEW
181 RCL 30
182 "  MOL W
   T WET = "
183 ARCL X
184 AVIEW
185 RCL 35
186 "  VELOC
   ITY FPS "
187 "⊢= "
188 ARCL X
189 AVIEW
190 RCL 37
191 "  STACK
    AREA = "
192 ARCL X
193 AVIEW
194 RCL 38
195 FIX 0
196 "  STACK
    ACFM = "
197 ARCL X
198 AVIEW
199 RCL 39
200 "* STACK
    DSCFM ="
201 "⊢ "
202 ARCL X
203 AVIEW
204 FIX 2
205 RCL 09
206 X=0?
207 GTO 03
208 .1558
209 *
210 RCL 28
211 *
212 RCL 39
213 *
214 "*OTHER
   LB/HR = "
215 "⊢ "
216 ARCL X
217 AVIEW

218◆LBL 03
219 RCL 40
220 "  % ISO
   KINETIC "
221 "⊢= "
222 "⊢ "
223 ARCL X
224 AVIEW
225 ADV
226 ADV
227 "END OF
   FIELD DA"
228 "⊢TA"
229 AVIEW
230 ADV
231 ADV
232 STOP

233◆LBL A
234 1
235 RCL 17
236 /
237 RCL 30
238 /
239 SQRT
240 RCL 32
241 *
242 RCL 31
243 *
244 85.49
245 *
246 STO 43
247 GTO B
248 .END.
```

STACK EMISSIONS DETERMINATION (MASSFLO)

The MASSFLO program calculates the emission rates for the front half, back half, and total catch. It requires four inputs: volume of the dry gas meter at standard conditions, stack dry standard cubic feet per minute, and the front and back milligrams of catch. If there is no back half catch, press R/S when the back half PROMPT appears and the calculator will determine the front half emissions and then stop. The back side of the Sampling Summary Sheet (Summary of Emissions) has spaces for emission results.

The emission values that the calculator provides are preceded by the letters F, B, and T. F designates Front Half, B is for Back Half, and T is for Total Emissions.

If the program is being run without the printer attached, the program can be stopped when displaying values. Since the label for each value is scrolled accross the display, pressing R/S when the equals (=) sign appears will stop the program and the numerical value will appear in the display.

The label for this program is MASSFLO. The Memory size is 024.

```
PRP "MASSFLO
      "

 01*LBL "MAS
       SFLO"
 02 ADV
 03 "RUN NUM
       BER"
 04 PROMPT
 05 ADV
 06 "VOL MTR
      STD ?"
 07 PROMPT
 08 STO 01
 09 "STACK D
      SCFM ?"
 10 PROMPT
 11 STO 02
 12 "FRONT 1
    /2 MG ?"
 13 PROMPT
 14 STO 03
 15 0
 16 "BACK 1/
      2 MG ?"
 17 PROMPT
 18 STO 04
 19 ADV
 20 ADV
 21 RCL 03
 22 64.8
 23 /
 24 RCL 01
 25 /
 26 STO 06
 27 RCL 04
 28 64.8
 29 /
 30 RCL 01
 31 /
 32 STO 10
 33 RCL 06
 34 +
 35 STO 20
 36 RCL 03
 37 RCL 01
 38 /
 39 35.314
 40 *
 41 STO 07
 42 RCL 04
 43 RCL 01
 44 /
 45 35.314
 46 *
 47 STO 17
 48 RCL 07
 49 +
 50 STO 21
 51 RCL 02
 52 RCL 06
 53 *
 54 60
 55 *
 56 7000
 57 /
 58 STO 08
 59 RCL 02
 60 RCL 10
 61 *
 62 60
 63 *
 64 7000
 65 /
 66 STO 18
 67 RCL 08
 68 +
 69 STO 22
 70 RCL 08
 71 .4536
 72 *
 73 STO 09
 74 RCL 18
 75 .4536
 76 *
 77 STO 19
 78 RCL 09
 79 +
 80 STO 23
 81 RCL 03
 82 RCL 04
 83 +
 84 STO 05
 85 RCL 06
 86 "F   GR/D
      SCF = "
 87 ARCL X
 88 AVIEW
 89 RCL 07
 90 "F   MG/M
       MM = "
 91 ARCL X
 92 AVIEW
 93 RCL 08
 94 "F   LB/H
        R = "
 95 ARCL X
 96 AVIEW
 97 RCL 09
 98 "F   KG/H
        R = "
 99 ARCL X
100 AVIEW
101 ADV
102 RCL 04
103 X=0?
104 STOP
105 RCL 10
106 "B   GR/D
      SCF = "
107 ARCL X
108 AVIEW
109 RCL 17
110 "B   MG/M
       MM = "
111 ARCL X
112 AVIEW
113 RCL 18
114 "B   LB/H
        R = "
115 ARCL X
116 AVIEW
117 RCL 19
118 "B   KG/H
        R = "
119 ARCL X
120 AVIEW
121 ADV
122 RCL 20
123 "T   GR/D
      SCF = "
124 ARCL X
125 AVIEW
126 RCL 21
127 "T   MG/M
       MM = "
128 ARCL X
129 AVIEW
130 RCL 22
131 "T   LB/H
        R = "
132 ARCL X
133 AVIEW
134 RCL 23
135 "T   KG/H
        R = "
136 ARCL X
137 AVIEW
138 END
```

VELOCITY TRAVERSE PROGRAM (METH 2)

The velocity traverse program determines the average stack velocity in feet per second and feet per minute. It gives the Actual Cubic Feet per Minute (ACFM) as well as Dry Standard Cubic Feet per Minute (DSCFM). If the stack is rectangular instead of round, simply press R/S when the stack diameter prompt appears and enter the stack area in square feet when the stack area prompt appears. If another gas is present in the stack effluent - such as a high concentration of SO_2, enter the molecular weight of the gas. The calculator will then prompt for PPM.

The label for this program is METH 2. The memory size is 045.

```
PRP "METH 2"

 01*LBL "MET
        H 2"
 02 CLRG
 03 "SITE ?"
 04 PROMPT
 05 0
 06 "STACK D
   IA INCH?"
 07 PROMPT
 08 STO 36
 09 X≠0?
 10 GTO 01
 11 "AREA SQ
       FT ?"
 12 PROMPT
 13 STO 37
 14 GTO 02

 15*LBL 01
 16 24
 17 /
 18 X↑2
 19 PI
 20 *
 21 STO 37

 22*LBL 02
 23 "NO TRAV
     PTS. ?"
 24 BEEP
 25 PROMPT
 26 STO 00
 27 STO 01
 28 "BAR PRE
       SS ?"
 29 PROMPT
 30 STO 05
 31 "STATIC
   IN HG ?"
 32 PROMPT
 33 STO 25
 34 13.6
 35 /
 36 RCL 05
 37 +
 38 STO 26
 39 "% MOIST
      URE ?"
 40 PROMPT
 41 STO 17
 42 100
 43 -
 44 CHS
 45 .01
 46 *
 47 STO 18
 48 "PITOT C
       P ?"
 49 PROMPT
 50 STO 27
 51 "% CO2 ?
          "
 52 PROMPT
 53 STO 19
 54 "% OXYGE
       N ?"
 55 PROMPT
 56 STO 20
 57 "% CO ?"
 58 PROMPT
 59 STO 22
 60 0
 61 "MOL WT
    OTHER ?"
 62 PROMPT
 63 .01
 64 *
 65 STO 41
 66 X=0?
 67 GTO A
 68 "PPM ?"
 69 PROMPT
 70 STO 42

 71*LBL A
 72 RCL 19
 73 RCL 20
 74 +
 75 RCL 22
 76 +
 77 RCL 42
 78 .0001
 79 *    .
 80 STO 43
 81 +
 82 100
 83 -
 84 CHS
 85 STO 21
 86 RCL 19
 87 .44
 88 *
 89 STO 23
 90 RCL 20
 91 .32
 92 *
 93 ST+ 23
 94 RCL 21
 95 RCL 22
 96 +
 97 .28
 98 *
 99 ST+ 23
100 RCL 43
101 RCL 41
102 *
103 ST+ 23
104 RCL 23
105 ADV
106 FIX 2
107 "MWd =  "
108 ARCL X
109 AVIEW
110 PSE
111 FIX 4
112 100
113 RCL 17
114 -
115 100
116 /
117 STO 18
118 1
119 RCL 18
120 -
121 18
122 *
123 RCL 23
124 RCL 18
125 *
126 +
127 STO 24
128 FIX 2
129 "MW WET
        =  "
130 ARCL X
131 AVIEW
132 STO 24
133 ADV
134 FIX 4

135*LBL 03
136 ADV
137 ADV
138 RCL 06
139 1
140 +
141 STO 06
142 FIX 0
143 TONE 7
144 "DELTA P
          "
145 ARCL X
146 AVIEW
147 STOP
148 ST+ 03
149 STO 04
150 "STACK T
      EMP?"
151 PROMPT
```

```
152 ST+ 30
153 460
154 +
155 *
156 RCL 24
157 /
158 RCL 26
159 /
160 SQRT
161 RCL 27
162 *
163 85.49
164 *
165 STO 04
166 Σ+
167 RCL 04
168 FIX 0
169 "FPS =  "
170 ARCL X
171 AVIEW
172 RCL 00
173 1
174 -
175 X=0?
176 GTO 04
177 STO 00
178 GTO 03

179*LBL 04
180 ADV
181 ADV
182 MEAN
183 "AVE FPS
          = "
184 ARCL X
185 AVIEW
186 60
187 *
188 "AVE FPM
          = "
189 ARCL X
190 AVIEW
191 FIX 2
192 STO 05
193 RCL 03
194 RCL 01
195 /
196 "AVE DEL
    TA P = "
197 ARCL X
198 AVIEW
199 RCL 26
200 "STK PRS
    . ABS = "
201 ARCL X
202 AVIEW
203 RCL 30
204 RCL 01
205 /
206 STO 44
207 FIX 0
208 "AVE STK
    TEMP = "
209 ARCL X
210 AVIEW
211 RCL 05
212 RCL 37
213 *
214 "STACK A
     CFM = "
215 ARCL X
216 AVIEW
217 RCL 26
218 *
219 528
220 *
221 29.92
222 /
223 RCL 44
224 460
225 +
226 /
227 RCL 18
228 *
229 "DSCFM =
          "
230 ARCL X
231 AVIEW
232 ADV
233 ADV
234 ADV
235 END
```

PROBE MARKING PROGRAM (METH 1)

The probe marking program locates the points at which the probe should be marked prior to sampling. The point locations given by the program include the nipple length and corrections are made for points that would otherwise be too close to the stack wall - that is the outermost points that would be closer than ½ inch to the stack wall for ducts 24 inches or less in diameter, and less than 1 inch from the stack wall for ducts greater than 24 inches in diameter.

The label for this program is METH 1. The Memory size is 036.

```
PRP "METH 1"

 01*LBL "MET
        H 1"
 02 ADV
 03 ADV
 04 CLRG
 05 "DIA INC
      HES?"
 06 24
 07 PROMPT
 08 STO 01
 09 X<=Y?
 10 XEQ a
 11 XEQ b

 12*LBL a
 13 .5
 14 STO 07
 15 GTO A

 16*LBL b
 17 1
 18 STO 07

 19*LBL A
 20 0
 21 "NIPPLE
     INCH ?"
 22 PROMPT
 23 STO 35
 24 "POINTS
   ONE TRW?"
 25 PROMPT
 26 STO 06
 27 1
 28 -
 29 STO 08
 30 RCL 06
 31 2
 32 /
 33 STO 02
 34 STO 00

 35*LBL 00
 36 RCL 00
 37 2
 38 *
 39 1
 40 -
 41 8
 42 RCL 02
 43 *
 44 /
 45 SQRT
 46 RCL 01
 47 *
 48 CHS
 49 RCL 01
 50 2
 51 /
 52 STO 03
 53 +
 54 RCL 07
 55 X>Y?
 56 XEQ c
 57 XEQ d

 58*LBL 02
 59 RCL 35
 60 +
 61 STO 04
 62 FIX 0
 63 RCL 05
 64 1
 65 +
 66 STO 05
 67 "POINT
                "
 68 ARCL X
 69 AVIEW
 70 PSE
 71 CLD
 72 FIX 1
 73 RCL 04
 74 TONE 8
 75 VIEW X
 76 PSE
 77 PSE
 78 1
 79 ST- 00
 80 RCL 00
 81 X=0?
 82 GTO 00
 83 1
 84 STO 00

 85*LBL 01
 86 RCL 00
 87 2
 88 *
 89 1
 90 -
 91 8
 92 RCL 02
 93 *
 94 /
 95 SQRT
 96 RCL 01
 97 *
 98 RCL 02
 99 +
100 STO 09
101 RCL 08
102 RCL 05
103 X=Y?
104 XEQ e
105 RCL 09
106*LBL 03
107 RCL 35
108 +
109 STO 04
110 RCL 05
111 FIX 0
112 1
113 +
114 STO 05
115 "POINT
                "
116 ARCL X
117 AVIEW
118 PSE
119 FIX 1
120 RCL 04
121 TONE 8
122 VIEW X
123 PSE
124 PSE
125 CLD
126 1
127 ST+ 00
128 RCL 06
129 RCL 05
130 X=Y?
131 STOP
132 GTO 01

133*LBL c
134 RCL 07
135 GTO 02
136 RTN

137*LBL d
138 RCL Y
139 RTN

140*LBL e
141 RCL 01
142 RCL 07
143 -
144 RCL 09
145 X<>Y
146 X<=Y?
147 GTO 03
148 RCL Y
149 GTO 03
150 .END.
```

NUMBER OF POINTS (NOP)

This program calculates the number of total sample points needed for a site based on the number of upstream and downstream duct diameters available and the duct diameter in inches. Prior to the change in the Federal Register in September 1983 that reduced the total number of sampling points, the program was very useful in determining the total number of sampling points at a site. The graph in the Federal Register has now been quite easy to read, so the program is probably more valuable as a means to check the number of points that the source tester determines from the Federal Register. The program is interesting to run, however, since it will point out errors if incorrect data should be entered. For example, if a duct diameter of less than 12 inches is entered, or if the number of duct diameters are too short, an error code will result.

The program label is NOP. The Memory size is 005.

```
  PRP "NOP"

01*LBL "NOP
         "
02 CLRG
03 FS? 55
04 ADV
05 8
06 "DOWNSTR
  EAM DIA?"
07 TONE 7
08 PROMPT
09 STO 00
10 X<=Y?
11 GTO 00
12 8
13 STO 00

14*LBL 00
15 2
16 RCL 00
17 X=Y?
18 GTO 01
19 X>Y?
20 GTO 01
21 "ERROR"
22 AVIEW
23 "DOWNSTR
   EAM DIA"
24 AVIEW
25 "TOO SHO
       RT"
26 AVIEW

27*LBL 02
28 TONE 7
29 GTO 02

30*LBL 01
31 2
32 "UPSTREA
    M DIA?"
33 TONE 7
34 PROMPT
35 STO 01
36 X<=Y?
37 GTO 03
38 2
39 STO 01

40*LBL 03
41 .5
42 RCL 01
43 X=Y?
44 GTO 04
45 X>Y?
46 GTO 04
47 "ERROR"
48 AVIEW
49 "UPSTREA
    M DIA"
50 AVIEW
51 "TOO SHO
       RT"
52 AVIEW
53 GTO 02

54*LBL 04
55 24
56 "STK DIA
   INCHES?"
57 TONE 7
58 PROMPT
59 STO 02
60 FS? 55
61 ADV
62 X>Y?
63 GTO 06
64 12
65 RCL 02
66 X<Y?
67 GTO 07
68 GTO 06

69*LBL 07
70 "ERROR"
71 AVIEW
72 "DIA < 1
     2 IN."
73 AVIEW
74 GTO 02

75*LBL 06
76 24
77 RCL 02
78 X>Y?
79 GTO a
80 RCL 00
81 8
82 X<=Y?
83 XEQ A

84*LBL a
85 RCL 00
86 7
87 X<=Y?
88 XEQ B
89 RCL 00
90 6
91 X<=Y?
92 XEQ C
93 RCL 00
94 5
95 X<=Y?
96 XEQ D
97 XEQ E

98*LBL "Y"
99 24
100 RCL 02
101 X>Y?
102 GTO b
103 RCL 01
104 2
105 X<=Y?
106 XEQ F

107*LBL b
108 RCL 01
109 1.75
110 X<=Y?
111 XEQ G
112 RCL 01
113 1.5
114 X<=Y?
115 XEQ H
116 RCL 01
117 1.25
118 X<=Y?
119 XEQ I
120 XEQ J

121*LBL "Z"
122 RCL 03
123 RCL 04
124 X<=Y?
125 XEQ "K"
126 XEQ "L"

127*LBL "K"
128 RCL 03
129 GTO "M"

130*LBL "L"
131 RCL 04

132*LBL "M"
133 FIX 0
134 "TOTAL P
   OINTS = "
135 ARCL X
136 AVIEW
137 FS? 55
138 ADV
139 FIX 4
140 FS? 55
141 ADV
142 FS? 55
143 ADV
144 FS? 55
```

```
145 ADV
146 BEEP
147 STOP

148*LBL A
149 8
150 STO 03
151 GTO "Y"

152*LBL B
153 12
154 STO 03
155 GTO "Y"

156*LBL C
157 16
158 STO 03
159 GTO "Y"

160*LBL D
161 20
162 STO 03
163 GTO "Y"

164*LBL E
165 24
166 STO 03
167 GTO "Y"

168*LBL F
169 8
170 STO 04
171 GTO "Z"

172*LBL G
173 12
174 STO 04
175 GTO "Z"

176*LBL H
177 16
178 STO 04
179 GTO "Z"

180*LBL I
181 20
182 STO 04
183 GTO "Z"

184*LBL J
185 24
186 STO 04
187 GTO "Z"
188 .END.
```

AVERAGE OF INDIVIDUAL DELTA P_s $[T_s + 460]^{\frac{1}{2}}$ (PRESS)

The PRESS program averages the individual square roots of the Delta P_s times the stack temperature + 460. It also averages the ^{o}F of the stack temperature. The program is used when summarizing the field data sheets.

The Federal Register does not require the above technique to be used in calculating the isokinetic sampling rate, but simply says that the average of all the Delta P_s values and the average of all the stack temperatures can be placed under the square root radical and extracted once for use in the calculations. In the majority of cases, it will make little difference which method is used, however, in a few stacks where there is a wide variation in the Delta P_s values it will be noticeable. The introduction of programmable hand calculators makes it just as easy to obtain the number correctly, and requires no more effort. If the Meter Box Program (DELTA H) is used for the sample run, the results of the PRESS program will already be calculated and there will be no need to use this program. This, of course, assumes that no errors were made entering the data in the DELTA H program and that no interruptions occurred in the test run.

The label for this program is PRESS. The Memory size is 004.

```
PRP "PRESS"

01◆LBL "PRE
       SS"
02 CLRG
03 "NO. OF
  POINTS ?"
04 PROMPT
05 STO 01
06 STO 03

07◆LBL 01
08 "DELTA P
       ?"
09 TONE 7
10 PROMPT
11 "STACK T
     EMP ?"
12 PROMPT
13 ST+ 02
14 460
15 +
16 *
17 SQRT
18 Σ+
19 RCL 01
20 1
21 -
22 STO 01
23 X≠0?
24 GTO 01
25 BEEP
26 FS? 55
27 ADV
28 MEAN
29 "SQRT PS
     TS = "
30 ARCL X
31 AVIEW
32 PSE
33 PSE
34 RCL 02
35 RCL 03
36 /
37 "AVE STK
   TEMP = "
38 ARCL X
39 AVIEW
40 END
```

EQUIVALENT DIAMETER (EDIA)

The Equivalent Diameter Program is useful on pre-test survey visits at sites where square or rectangular ducts exist. The dimensions of the duct are measured in inches and then enter into the program as prompted for. The program then determines the equivalent diameter of the duct in both feet and inches. Once the equivalent diameter of the duct is determined, it is possible to decide if there is enough straight run of duct to provide an adequate sampling site. Since the minimum number of duct diameters for a site is $2\frac{1}{2}$, simply multiply the equivalent diameter by $2\frac{1}{2}$ to see if the site meets the minimum requirements of Method 1.

The label for this program is EDIA. The Memory size is 003.

```
PRP "EDIA"

01♦LBL "EDI
       A"
02 FIX 1
03 "LONG IN
     CH?"
04 PROMPT
05 STO 00
06 "SHORT I
     NCH?"
07 PROMPT
08 STO 01
09 ADV
10 ADV
11 RCL 00
12 *
13 2
14 *
15 STO 02
16 RCL 00
17 RCL 01
18 +
19 RCL 02
20 X<>Y
21 /
22 12
23 /
24 "EQUIV D
  IA FT = "
25 ARCL X
26 AVIEW
27 ADV
28 12
29 *
30 "EQUIV D
  IA INCHE"
31 "⊢S = "
32 ARCL X
33 AVIEW
34 ADV
35 ADV
36 BEEP
37 END
```

SAMPLE POINT TIMES USING THE ALIGNMENT METHOD (TIMES)

This program calculates the individual sample times in minutes and seconds for sample points where the Alignment Method is used. The program requires only three inputs: first, there is the normal sample time per point if the Alignment Method were not being used. Second, there is the total number of points to be sampled. Third, the angle of maximum flow at these points must be entered. After these data have been entered, the calculator provides the sample time per point in minutes and seconds. A decimal separates the minutes and seconds. A time of 5 minutes and 14 seconds will appear in the display as 5.14 and will be printed in the same manner on the printer tape if the printer is used. The program, of course, will run with or without the printer, and executing the program without the printer will be discussed first. NOTE: The first five steps of the instructions are the same for use with or without the printer.

1. Engage the program by pressing XEQ ALPHA TIMES ALPHA. NO OF POINTS? will appear in the display.
2. Enter the total number of sample points for the test and press R/S.
3. NORMAL TIME? will appear in the display. Enter the sample time per point in minutes that would normally be used if the Alignment Method were not being used. Press R/S.
4. ANGLE 1. will appear in the display. Enter angle 1 in degrees and press R/S.
5. ANGLE 2. will appear in the display. Enter angle 2 in degrees and press R/S. Continue in this manner until all angles have been entered.
6. When all the angles have been entered, the calculator will display the average angle, and pause long enough for it to be written down. Then the decimal base time will be displayed. The calculator will pause long enough for this number to be recorded also.
7. WORKING will next appear in the display. The time for each sample point in minutes and seconds is now being determined.
8. The next message to appear in the display will be: MIN AND SEC FOR EACH POINT FOLLOW.
9. The calculator will now begin displaying the minutes and seconds of sample time for each point. An audible tone will occur, the display will

briefly show the point number and then give the minutes and seconds for that point. The program stops at each point so that the time can be recorded. Pressing R/S moves on to the next point until all the points have been displayed.

• USING THE PROGRAM WITH THE PRINTER

The program runs much the same with the printer as it does without the printer. Complete steps 1 through 5 on the previous page.

6. After all the angles have been entered, the display will say EDIT. Then the memories that contain angles will be printed on the calculator tape. Angle 1 is stored in memory 1 (or Register 1) and will be displayed as R01 = nn. The actual angles can be checked against the corresponding memories and corrected if necessary. To correct an angle, simply store the correct value in the appropriate memory register.
7. After reviewing the angles for correctness, press R/S. The average angle and the decimal base time will be displayed and printed. The display will now show WORKING. The calculator is now determining the sample times in minutes and seconds for each point.
8. A message saying MIN AND SEC FOR EACH POINT FOLLOW will appear in the display and on the tape. Then the printer will provide a list of memories that give the time in minutes and seconds for each sample point. Thus, R01= 5.25 would indicate that sample point 1 should be sampled for 5 minutes and 25 seconds. The register number (or memory number) is the same as the point number.

When using the printer, there are three mode positions in which the printer can be operated. They are MAN (for manual), TRACE, and NORM (for normal). Since the TRACE mode is used to find errors in programs, it need not be considered here. This leaves the MAN and NORM modes, and either may be used when running the program. The MAN mode is the most efficient since it prints only when instructed to and therefore saves printer tape. The NORM mode prints all entries and uses more tape, but it does provide a copy of data input.

The label for this program is TIMES. The Memory size is 110

```
PRP "TIMES"

01◆LBL "TIM
        ES"
02 CF 12
03 CLRG
04 "NO OF P
   OINTS ?"
05 PROMPT
06 STO 26
07 1
08 STO 27
09 "NORMAL
    TIME ?"
10 PROMPT
11 STO 28
12 1
13 RCL 26
14 .001
15 *
16 +
17 STO 00
18 STO 29

19◆LBL 00
20 RCL 27
21 FIX 0
22 TONE 7
23 "ANGLE "
24 ARCL X
25 PROMPT
26 ST+ 30
27 STO IND
          00
28 ISG 00
29 1
30 ST+ 27
31 RCL 26
32 RCL 27
33 X>Y?
34 GTO 01
35 GTO 00

36◆LBL 01
37 FS? 55
38 GTO 02
39 GTO 03

40◆LBL 02
41 "EDIT"
42 AVIEW
43 RCL 29
44 PRREGX
45 STOP

46◆LBL 03
47 FIX 2
48 RCL 30
49 RCL 26
50 /
51 "AVE ANG
     LE = "
52 ARCL X
53 AVIEW
54 PSE
55 STO 31
56 COS
57 RCL 28
58 X<>Y
59 /
60 "DECIMAL
          "
61 AVIEW
62 PSE
63 "BASE TI
     ME = "
64 ARCL X
65 AVIEW
66 PSE
67 STO 32
68 "PLEASE
      WAIT"
69 AVIEW
70 RCL 29
71 STO 00
72 STO 33

73◆LBL 04
74 RCL IND
          00
75 ISG 00
76 GTO 05

77◆LBL 05
78 COS
79 RCL 32
80 *
81 60
82 /
83 HMS
84 .01
85 /
86 STO IND
          33
87 ISG 33
88 GTO 04
89 "MIN AND
   SEC FOR"
90 AVIEW
91 "EACH PO
   INT FOLL"
92 "OW"
93 AVIEW
94 FC? 55
95 GTO 07
96 CF 13
97 SF 12
98 RCL 29
99 PRREGX
100 STOP

101◆LBL 07
102 RCL 29
103 STO 00
104 0
105 STO 27

106◆LBL 08
107 1
108 ST+ 27
109 RCL 27
110 FIX 0
111 "POINT "
112 ARCL X
113 AVIEW
114 TONE 8
115 PSE
116 FIX 2
117 RCL IND
          00
118 ISG 00
119 GTO 09

120◆LBL 09
121 VIEW X
122 STOP
123 GTO 08
124 END
```

TIME WEIGHTED AVERAGE (TWA)

This program determines time-weighted averages that must be used to calculate the isokinetic sampling rate when the Alignment Method is used. The program is simple to use and prompts for data input. It can also be used for time-weighted averages for normal Method 5 sampling where time weighted averages are necessary for data reduction. This will be discussed at the end of this section.

This program differs from most of the programs in this booklet in that the memory size for this program is 110. To execute the program, do the following:

1. To engage the program, press XEQ ALPHA TWA ALPHA.
2. The NO. POINTS prompt will appear in the display. Enter the total number of sample points.
3. An audible tone will be heard and the point number will momentarily appear on the display. Then, the MINUTES ? prompt will appear. Enter the number of whole minutes sampled at that point. (For a point with a sample time of 5 min. and 27 sec. enter "5".) Press R/S.
4. The SECONDS ? prompt will appear in the display. Enter the seconds and press R/S. (For the above example, enter 27 and press R/S.)
5. Enter minutes and seconds for all points.
6. The total seconds for the sampling period will appear briefly in the display/ A message saying BEGIN DELTA H will appear briefly in the display followed by DELTA H 1. Enter the first Delta H and press R/S. Enter all Delta H values.
7. The next message in the display will be BEGIN MTR TEMP. This will be followed by the TEMP IN? prompt. Enter the inlet dry gas meter temperature and press R/S. The TEMP OUT? prompt will then appear. Enter the outlet temperature and press R/S. Enter all inlet and outlet temperatures. NOTE: If you have a single dry gas meter temperature, set Flag 00. This will prompt for one meter temperature instead of two.
8. After the meter temperatures are entered, a message saying BEGIN STK TEMP will appear in the display followed by STK TEMP 1. Enter all stack temperatures in ^{o}F.
9. After all stack temperatures are entered, a message saying BEGIN SQRT PSTS will appear. This will be followed by DELTA P 1. Enter all Delta P values.

10. After all Delta P values have been entered, a message will appear briefly saying BEGIN COS SQRT PSTS. This will be followed with DEGREES ∠ 1. Enter the angle in degrees of all sampling points.
11. At this point, the calculator will begin giving the time weighted averages for data previously entered. If no printer is being used, the first average will be displayed and the program will stop. Simply press R/S each time a new average is desired. The summaries from this program can be used with the METH 5 program to obtain the isokinetic sample rate and volumetric flow rates.

To use this program to obtain time-weighted averages for for runs where the Alignment Method was not used, begin by entering the minutes and seconds. If there is a point with an even number of minutes and no seconds, enter a 0 for the seconds. Continue to enter data as prompted for until the message: BEGIN COS SQRT PSTS. At this point, the degrees for the angles are entered if the run used the Alignment Method. Since this would not be an Alignment Method run, enter a 0 for the degrees of each angle (Cos $0^{\circ} = 1$). After zeros have been entered for all the angles, the calculator will then provide the necessary time-weighted averages.

The label for this program is TWA. The Memory size is 110.

```
   PRP "TWA"

01*LBL "TWA
       "
02 CLRG
03 1
04 STO 00
05 "NO. POI
     NTS ?"
06 PROMPT
07 STO 50
08 .001
09 *
10 ST+ 00
11 RCL 00
12 STO 52

13*LBL 01
14 1
15 ST+ 49
16 RCL 49
17 FIX 0
18 "POINT "
19 ARCL X
20 AVIEW
21 TONE 8
22 "MINUTES
         ?"
23 PROMPT
24 60
25 *
26 "SECONDS
         ?"
27 PROMPT
28 +
29 ST+ 51
30 STO IND
        00
31 ISG 00
32 GTO 01
33 RCL 52
34 STO 00
35 STO 53

36*LBL 02
37 RCL IND
        00
38 ISG 00
39 GTO 03

40*LBL 03
41 RCL 51
42 /
43 STO IND
        53
44 ISG 53
45 GTO 02
46 RCL 52
47 STO 00
48 RCL 51
49 "TOTAL S
     EC = "
50 ARCL X
51 AVIEW
52 BEEP
53 "BEGIN D
     ELTA H"
54 AVIEW
55 0
56 STO 49

57*LBL 04
58 1
59 ST+ 49
60 RCL 49
61 "DELTA H
       "
62 ARCL X
63 TONE 8
64 PROMPT
65 RCL IND
        00
66 *
67 ST+ 54
68 ISG 00
69 GTO 04
70 RCL 52
71 STO 00
72 BEEP
73 "BEGIN M
   TR TEMP"
74 AVIEW

75*LBL 05
76 FS? 00
77 GTO A
78 GTO B

79*LBL A
80 "METER T
     EMP?"
81 TONE 8
82 PROMPT
83 GTO 14

84*LBL B
85 "TEMP IN
       ?"
86 TONE 7
87 PROMPT
88 "TEMP OU
       T?"
89 TONE 8
90 PROMPT
91 +
92 2
93 /

94*LBL 14
95 RCL IND
        00
96 *
97 ST+ 55
98 ISG 00
99 GTO 05
100 RCL 52
101 STO 00
102 RCL 50
103 61
104 +
105 .001
106 *
107 62
108 +
109 STO 60
110 STO 59
111 BEEP
112 "BEGIN S
    TK TEMP"
113 AVIEW
114 0
115 STO 49

116*LBL 06
117 1
118 ST+ 49
119 RCL 49
120 "STK TEM
       P "
121 ARCL X
122 TONE 7
123 PROMPT
124 STO IND
        60
125 ISG 60
126 GTO 07

127*LBL 07
128 RCL IND
        00
129 *
130 ST+ 56
131 ISG 00
132 GTO 06
133 RCL 52
134 STO 00
135 RCL 59
136 STO 60
137 STO 61
138 BEEP
139 "BEGIN S
    QRT PSTS"
```

```
140 AVIEW
141 0
142 STO 49

143*LBL 08
144 1
145 ST+ 49
146 RCL 49
147 TONE 8
148 "DELTA P
      "
149 ARCL X
150 PROMPT
151 RCL IND
      60
152 460
153 +
154 *
155 SQRT
156 ISG 60
157 GTO 09

158*LBL 09
159 RCL IND
      00
160 *
161 ST+ 57
162 ISG 00
163 GTO 11

164*LBL 11
165 STO IND
      61
166 ISG 61
167 GTO 08
168 RCL 59
169 STO 60
170 BEEP
171 "BEGIN C
   OS SQRT "
172 "⊢ PSTS"
173 AVIEW
174 0
175 STO 49

176*LBL 12
177 1
178 ST+ 49
179 RCL 49
180 "DEGREES
      ∠ "
181 ARCL X
182 TONE 7
183 PROMPT
184 COS
185 RCL IND
      60
186 *
187 ST+ 58
188 ISG 60
189 GTO 12
190 RCL 54
191 FIX 2
192 "DELTA H
      = "
193 ARCL X
194 AVIEW
195 FC? 55
196 STOP
197 RCL 55
198 FIX 0
199 "METER T
    EMP = "
200 ARCL X
201 AVIEW
202 FC? 55
203 STOP
204 RCL 56
205 "STACK T
    EMP = "
206 ARCL X
207 AVIEW
208 FC? 55
209 STOP
210 RCL 57
211 FIX 4
212 "SQRT PS
     TS = "
213 ARCL X
214 AVIEW
215 FC? 55
216 STOP
217 RCL 58
218 "COS SQR
  T PSTS ="
219 "⊢ "
220 ARCL X
221 AVIEW
222 FC? 55
223 STOP
224 FIX 2
225 RCL 51
226 60
227 /
228 "MINUTES
      = "
229 ARCL X
230 AVIEW
231 .END.
```

SULFUR DIOXIDE EMISSIONS (METH 6)

This program calculates the pounds per dry standard cubic foot of SO_2 emissions, and it also calculates the emission rate in pounds of SO_2 produced for each one million BTUs of heat input. The name of the program is METH 6, and the memory size is 019.

The way to use the program is to first assign it to a user key. With the calculator in the user mode, pressing the assigned key will engage the program. The program will then prompt for data input. The first prompt will be: METH 5 OR 6 EQUIP? Enter either a 5 or a 6. Then continue to enter data as prompted for.

When doing Method 6 analyses, the majority of values needed to reach the answers remain the same. If doing several runs, the meter volume, temperature of the meter, and volume of titrant will vary from run to run. If Method 5 equipment is used in obtaining the sample, the Delta H will also be a variable. After the program has been run one time, the values that remain constant will not be prompted for again. Thus, after finding the first emission rate, pressing R/S will cause the program to reset itself and the first prompt will be DELTA H? if using Method 5 equipment, or METER VOL? if using Method 6 equipment. The second analysis determination can be made with only four or five prompts, depending upon the type of equipment used.

During the course of several analyses, some of the values that were prompted for initially may change. The easiest way to remedy this situation is to press the assigned USER key and start the program all over again. All the memories will be cleared and the new data can be stored. Another way (though not recommended) is to refer to the METHOD 6 SUMMARY SHEET. Above the abbreviation for each column value is a number. Most of the numbers are inside a circle. The number in the circle is the memory number in which that value is stored. Therefore, if the per cent oxygen changed, the new value could be manually stored in memory 17. While manually storing numbers will work, the most reliable way to change values is simply to run the program from the beginning by pressing the assigned USER key.

The label for this program is METH 6. The Memory size is 019.

```
PRP "METH 6"

 01*LBL "MET
        H 6"
 02 CLRG
 03 CF 00
 04 "METH 5
   OR 6 EQU"
 05 "HIP ?"
 06 PROMPT
 07 STO 18
 08 6
 09 X=Y?
 10 GTO 01

 11*LBL 02
 12 FS? 00
 13 GTO 04
 14 1
 15 "METER B
      OX Y?"
 16 PROMPT
 17 STO 03

 18*LBL 04
 19 0
 20 "DELTA H
           ?"
 21 PROMPT
 22 STO 04
 23 GTO 03

 24*LBL 01
 25 1
 26 FS? 00
 27 GTO 05
 28 "METER B
      OX Y?"
 29 PROMPT
 30 STO 03

 31*LBL 05
 32*LBL 03
 33 FS? 00
 34 GTO 06
 35 "BAR PRE
       SS ?"
 36 PROMPT
 37 STO 05

 38*LBL 06
 39 "METER V
       OL ?"
 40 PROMPT
 41 STO 06
 42 "TEMP MT
      R F. ?"
 43 PROMPT
 44 460
 45 +
 46 STO 07
 47 RCL 04
 48 13.6
 49 /
 50 RCL 05
 51 +
 52 RCL 06
 53 *
 54 17.64
 55 *
 56 RCL 07
 57 /
 58 RCL 03
 59 *
 60 FIX 4
 61 "VOL MTR
      STD = "
 62 ARCL X
 63 AVIEW
 64 PSE
 65 PSE
 66 STO 08
 67 FS? 00
 68 GTO 07
 69 "NORMALI
       TY ?"
 70 PROMPT
 71 STO 09

 72*LBL 07
 73 "ML TITR
      ANT?"
 74 PROMPT
 75 STO 10
 76 FS? 00
 77 GTO 08
 78 0
 79 "ML BLAN
       K ?"
 80 PROMPT
 81 STO 11

 82*LBL 08
 83 FS? 00
 84 GTO 09
 85 "ML SOLU
      TION?"
 86 PROMPT
 87 STO 12
 88*LBL 09
 89 FS? 00
 90 GTO 10
 91 "ML ALLI
      QUOT?"
 92 PROMPT
 93 STO 13

 94*LBL 10
 95 RCL 12
 96 RCL 13
 97 /
 98 STO 14
 99 RCL 10
100 RCL 11
101 -
102 RCL 14
103 *
104 RCL 09
105 *
106 RCL 08
107 /
108 0.000070
          6
109 *
110 SCI 2
111 "SO2 LB/
     DSCF = "
112 ARCL X
113 AVIEW
114 PSE
115 PSE
116 STO 15
117 FS? 00
118 GTO 11
119 "F FACTO
       R ?"
120 PROMPT
121 STO 16

122*LBL 11
123 FS? 00
124 GTO 12
125 "% OXYGE
       N ?"
126 PROMPT
127 STO 17

128*LBL 12
129 RCL 17
130 20.9
131 "GOOD LU
       CK"
```

```
132 AVIEW
133 X<>Y
134 -
135 20.9
136 X<>Y
137 /
138 RCL 15
139 *
140 RCL 16
141 *
142 FIX 4
143 "LB/10+6
     BTU = "
144 ARCL X
145 AVIEW
146 STOP
147 SF 00
148 RCL 18
149 6
150 X=Y?
151 GTO 03
152 GTO 02
153 .END.
```

METHOD 6 SUMMARY SHEET

PLANT _____ LOCATION _____

SAMPLED SOURCE _____

Run	Date	(3) Y_m	(4) ΔH	(5) P Bar.	(6) V_m	(7) T_m	8 $V_{m\,Std.}$	(9) N	(10) V_t	(11) V_{tb}	(12) V_{soln}	(13) V_a	15 LB/DSCF	(16) F	(17) $\%\omega_2$	E

Y_m = Meter Box Correction Factor

ΔH = Average Orfice Pressure Drop, Inches H_2O

P Bar. = Barometric Pressure, Inches Hg, Absolute

V_m = Volume of Dry Gas at Meter Conditions, DCF

T_m = Average Meter Temperature, ^oF.

$V_{m\,Std.}$ = Volume of Dry Gas at STP, DSCF[a]

N = Normality of Titrant

V_t = Volume of Titrant (ml.)

V_{tb} = Volume of Titrant Blank (ml.)

V_{soln} = Volume of Solution (ml.)

V_a = Volume of Aliquot (ml.)

LB/DSCF = Pounds per Dry Standard Cubic Foot Emission Rate

F = F Factor

$\%\omega_2$ = Per Cent Oxygen

E = Emission Rate in Pounds of Pollutant per One Million BTU Heat Input.

[a]Dry Standard Cubic Feet @ 68^oF, 29.92 in. Hg.

FLAGS

The chart below gives the Flags used in the programs and program name for which the Flag is used.

FLAG #	PROGRAM LABEL	FUNCTION
00	NOZZLE	Flag is set to choose a sample rate other than 0.75 DSCFM.
00	METH 3	Flag is set when determining wet and dry molecular weights of stack gas only, i.e., when running the METH 3 program alone.
01	METH 3	Set this Flag plus Flags 00 and 02 to get dry molecular weight only when running METH 3 program only.
02	METH 3	Same as above.
03	METH 4	Flag is set to prevent calculator from prompting for vapor pressure of water at stack temperature.
04	METH 4	Flag is set to prevent calculator from prompting for other gas removed before the dry gas meter.
05	DELTA H	This Flag is set if the meter box has only one indicator for temperature. This is done automatically.
06	METH 5	This Flag is set if the pitot coefficient is other than 0.84.
07	METH 5	This Flag is set if the Alignment Method was used in obtaining the sample.
08	METH 3	This Flag can be set to eliminate the CO prompt from the Method 3 program assuming no CO is present in the stack gas, or if a Fyrite is used and CO is not determined.

INSTRUCTIONS FOR THE HP 41

The instructions that follow in this section briefly outline the operation of the HP 41 for use with the programs contained in this booklet. The following areas will be discussed:

Calculator Models
The Keyboard
The RPN System
How to Manually Enter a Program
How to Enter a Program Using the Card Reader
How to Run a Program
How to Assign a Program to a User Key
How to Use the Catalogue Function
How to Use the Printer

CALCULATOR MODELS

There are presently three models of the HP 41 calculator that these programs can be used with. The models are the HP 41-C, the HP 41-CV, and the HP41-CX. The newer calculators, the 41-CV and 41-CX are acceptable as is. The older model HP-41C will require either an 82170A QUAD module, or four of the 82106-A memory modules.

THE CALCULATOR KEYBOARD

The description of the keyboard is for those individuals that are unfamiliar with the HP 41 calculators. Only a few of the keys are discussed and those familiar with the keyboard may wish to skip this discussion.

Immediately below the display of the HP 41 are the toggle keys. Reading from left to right, the toggle keys are ON USER and PRGM ALPHA. The ON key turns the calculator on and off. The USER toggle key enters the USER mode where either a program or a function can be assigned to a key (the USER mode will be discussed later.) The PRGM toggle key is used to enter the Program mode for either loading, writing, or correcting a program. The ALPHA mode is used for entering letters of the alphabet into the display.

There are some other keys used in running the programs in this booklet that are important. They are the Shift key, the Execute key, the Enter key, the Clear key, and the Run/Stop key.

The yellow key on the left side of the keyboard is the Shift key. This key is used to get to the functions written in yellow above the main keys.

There are some other keys used in running the programs in this booklet that are important. They are the Shift key, the Execute key, the Enter key, The Clear key, and the Run/Stop key.

The yellow key on the left side of the keyboard is the Shift Key. This key is used to get to the functions written in yellow above the main keys.

To the right of the yellow Shift key is the Execute (XEQ) key. This key is used when executing a program or function.

Below the Shift and Execute keys is the ENTER key. This key is used to enter numbers into the display when manually operating the calculator and will be discussed in the RPN section that follows.

Three keys from the right of the ENTER key is a key with an arrow pointing to the left. This key is used for clearing the display.

The bottom right-hand key is the Run/Stop key. This key is pressed to either start or stop a program.

THE RPN SYSTEM

This discussion is for those who have never used a calculator that uses the RPN system, or for those who may have limited experience with HP calculators or the HP 41. Those who are familiar with the RPN system may want to skip this part.

The HP 41 uses the RPN system or Reverse Polish Notation as it is also called. This logic is based on a mathametical logic known as "Polish Notation" developed by a noted Polish logician, Jan Lukasiewicz (1878 - 1956). Conventional algebraic logic places operators between relevant numbers (2 times 3 for example) whereas Lukasiewicz's notation specifies the operators before the variables. A variation of this logic that places the operators after the variables is called Reverse Polish Notation (2 ENTER, 3 times for example). Hewlett-Packard has choosen the RPN system for greater efficiency in digital computations.

When using the RPN system to perform computations, the ENTER key is usually pressed to enter the first number. In multiplying 2 times 3, the 2 key is pressed and then the ENTER key. Next, the 3 is pressed followed by the times key. The answer 6 appears in the display. If at this point, you wish to get the square root of 6, it is not necessary to press the ENTER key again; simply press the square root key and the answer will appear in the display. With a little practice and referral to the HP Owner's Manual, using the RPN system will become quite easy.

HOW TO MANUALLY ENTER A PROGRAM

A program can be manually entered into the calculator; however, it is somewhat time consuming. For this reason, a card reader is recommended. Although the card reader is an additional expenditure, one card reader can be used to program several calculators.

Before manually keying in a program, the memory size of the calculator must be set. This needs to be done only once before the first program is entered. To set the memory size, press the XEQ key, followed by the ALPHA toggle key. Then spell SIZE and press the ALPHA toggle key again. At this point, SIZE will appear in the display followed by three blank spaces. The memory size may now be entered. Most of the programs will run with a memory size of 045 or less. The Alignment Method programs require a memory size of 110. Unless the Alignment programs are going to be used, a memory size of 045 is recommended.

To manually enter a program, press the PRGM toggle key. Then press the Shift key followed by the GTO key. Next, press the decimal key twice. PACKING should appear briefly in the display.

The next step is to enter the program label. To do this, press the Shift key followed by the LBL key. Now press the ALPHA toggle key in order to spell the program label and press ALPHA again. From this point on, enter the program as it printed in the booklet. The last line in the program is another PACKING function, that is, the shift key, the GTO key, and the decimal key twice. The PACKING function places an END statement at the bottom of the program. Before loading another program, the PACKING function should be executed again.

While it is time consuming to enter programs manually, there is also another problem. The calculator will not hold all the programs in the booklet at one time. This means that some programs may have to be cleared so that others may be entered. If a card reader is not available, then the most frequently used programs could be manually entered and left in the calculator memory. For the purpose of checking the isokinetic sampling rate, the METH 3, METH 4, METH 5, and PRESS programs would be the programs to store.

HOW TO ENTER A PROGRAM USING THE CARD READER

Entering a program into the calculator memory using a card reader is very simple and fast. First, the memory size must be set as discussed in the preceeding section. Next, the PACKING function should be executed by pressing the Shift key,

the GTO key, and the decimal key twice. The program cards may now be processed through the card reader. The card is placed into the slot on the right hand side of the card reader. If the program uses the whole card, or more than one card, the calculator will prompt for additional input. The display will say RDY 02 OF 04, or something similar since the order in which the cards are entered makes no difference. If the display prompts for additional input, take the previously run card and turn it upside down and run it through the card reader again. When the last track (each card has two tracks) has been entered, WORKING will appear in the display. At this point, the entire program has been placed into the calculator memory.

Before entering another program, it is important to execute the PACKING function. The PACKING function moves the previously entered program from the bottom of the program memory towards the top of the memory, making it possible to enter the next program without writing over part of the previously entered program.

HOW TO RUN A PROGRAM

Running a program is very simple. Simply press the XEQ key, the ALPHA toggle key, spell the program label, and press the ALPHA toggle key again. The program will be engaged and the first prompt will appear. Simply enter the data prompted for and press the R/S key. After all the data have been entered, the calculator will then provide the answers.

Some of the programs begin with the prompt RUN NUMBER? If using a printer, the Run Number may be entered by using the ALPHA mode. Simply press the ALPHA toggle key, spell the run number, and press the ALPHA toggle key again. The run number will be printed at the top of the printer tape.

HOW TO ASSIGN A PROGRAM TO A USER KEY

Assigning a program to a USER key is quite easy. Simply press the Shift key, the ASN key and the ALPHA toggle key. ASN _ will appear in the display. Key in the program label and press the ALPHA toggle key again. A dash will appear to the right of the program label. Pressing one of the calculator keys will assign the program to that key. Whenever you want to run the program that was just assigned to the user key, simply press the USER toggle key and the key to which the program was assigned. The program will be engaged and the first prompt will appear.

A USER key is simply an easier way of calling up or executing a program.

HOW TO USE THE CATALOG FUNCTION

The CATALOG function lists the programs stored in the calculator memory. To execute this function, press the Shift key, the CATALOG key, and the "1" key. The calculator will begin listing the programs in its memory. The program label(s) will appear in the display followed by an END statement. The CATALOG function is a handy way to get to the beginning of a program for programming purposes. By pressing R/S at the appropriate time, the beginning of the program may be reached. Pressing the PRGM toggle key will then display the first line of the program.

HOW TO USE THE PRINTER

The printer plugs into one of the ports on the HP 41 (consult the Owner's Manual for the proper port). The printer has three modes: these are MAN (manual), TRACE, and NORM (normal). The TRACE mode is useful when trying to find errors in programs since it prints every operation the calculator performs while running that program. The NORM mode is most frequently used. It prints all the prompts found in a program and it also prints all the answers. The MAN mode prints only when instructed to by the program. This mode is useful when running the Alignment Method programs since it conserves printer tape.

HOW TO DELETE A PROGRAM

To delete or clear a program from the calculator memory, first press the XEQ key. Next, press the ALPHA toggle key and enter CLP into the display. Then press the ALPHA toggle key again. The display will show CLP_. Now, press the ALPHA toggle key again and spell the program label. After the program label has been entered into the display. press the ALPHA toggle key again. PACKING will appear in the display. When PACKING disappears from the display, the current X register value will appear and the program has been cleared from the calculator's memory.